

Modellazione di sistemi elettromeccanici con Lagrange

Ing. Dipl. ETH Roberto Bucher

7 febbraio 2011

1 Introduzione

Questo testo non vuole essere una introduzione al formalismo di Lagrange, ma un compendio pratico che presenta tutta una serie di sistemi modellati con questo sistema. I modelli sono poi sempre confrontati con un modello sviluppato tramite Modelica.

2 Il metodo di Lagrange

La metodologia di Lagrange richiede l'utilizzo di coordinate indipendenti q_i (coordinate generalizzate) e utilizza la descrizione energetica del sistema per determinare le equazioni differenziali che lo descrivono dinamicamente. Le grandezze in gioco sono:

U che rappresenta l'energia potenziale associata al sistema.

T che rappresenta l'energia cinetica associata al sistema.

D che rappresenta l'energia dissipata dal sistema.

L che rappresenta la "lagrangiana" del sistema calcolata come $L = T - U$.

Una volta trovata la "lagrangiana associata al sistema si possono determinare le equazioni differenziali associate al sistema con

$$\frac{d}{dt} \frac{\delta L}{\delta \dot{q}_i} - \frac{\delta L}{\delta q_i} + \frac{\delta D}{\delta \dot{q}_i} = F_{ext}$$

3 Calcolo dell'energia nei sistemi elettromeccanici

3.1 Sistemi meccanici

La grandezza generalizzata in un sistema meccanico è una posizione (x o φ).

3.2 Energia potenziale in un sistema meccanico

In un sistema meccanico l'energia potenziale può avere due forme

3.2.1 Energia potenziale dovuta alla gravità

$$U = m \cdot g \cdot h(q)$$

3.2.2 Energia potenziale accumulata in una molla

$$U = \frac{1}{2} \cdot k \cdot q^2$$

dove k rappresenta la costante della molla.

3.3 Energia cinetica in un sistema meccanico

$$T = \frac{1}{2} \cdot m \cdot \dot{q}_i^2$$

3.4 Energia dissipata in un sistema meccanico

$$D = \frac{1}{2} \cdot f_v \cdot \dot{q}_i^2$$

dove f_v rappresenta il coefficiente di attrito viscoso.

3.5 Energia potenziale in un sistema elettrico

3.6 Sistemi elettrici

La grandezza generalizzata utilizzata genericamente è la quantità di carica Q_i .

3.7 Energia potenziale in un sistema elettrico

È l'energia accumulata nei condensatori e vale

$$U = \frac{1}{2 \cdot C} \cdot Q_i^2$$

3.8 Energia cinetica in un sistema elettrico

È l'energia di una bobina e vale

$$T = \frac{1}{2} \cdot L \cdot \dot{Q}_i^2$$

3.9 Energia dissipata in un sistema elettrico

È l'energia dissipata dalla resistenza

$$D = \frac{1}{2} \cdot R \cdot \dot{Q}_i^2$$

4 Esempi pratici

4.1 Macchina di Atwood semplificata

La figura 1 rappresenta la macchina di Atwood. In questo primo esempio non consideriamo l'inerzia e l'attrito della parte rotante.

Le grandezze generalizzate sono qui la posizione $x_1(t) = x(t)$ della massa M_1 . Tramite la lunghezza della corda l possiamo determinare la posizione della seconda massa come $x_2(t) = l - x(t)$.

Le grandezze per il calcolo della lagrangiana sono:

$$T = \frac{1}{2} \cdot M_1 \cdot \dot{x}_1^2 + \frac{1}{2} \cdot M_2 \cdot \dot{x}_2^2$$




Figura 1: Macchina di Atwood

e

$$U = M_1 \cdot g \cdot x_1 + M_2 \cdot g \cdot x_2$$

Nel calcolo utilizziamo per la seconda massa la coordinata $(l - x(t))$.

Lo script per Maxima è il seguente:

```
x: x(t);
x1: x;
x2: (l-x);

dx: diff(x, t);
ddx: diff(x, t, 2);

dx1: diff(x1, t);
dx2: diff(x2, t);

T: 1/2*M1*dx1^2+1/2*M2*dx2^2;
U: M1*g*x1+M2*g*x2;
L: T-U;

eqn: diff(diff(L, dx), t)-diff(L, x)=0;

sol: linsolve(eqn, ddx);
```

$$\frac{d^2}{dt^2} x(t) = \frac{g M_2 - g M_1}{M_2 + M_1}$$

4.2 Macchina di Atwood completa

In questo caso consideriamo anche la parte rotante. La ruota ha raggio r e inerzia J . Inoltre è presente un attrito di rotazione f_v .

La coordinata generalizzata resta $x(t)$ ma nel calcolo, per semplicità viene introdotta una nuova coordinata $\varphi = x(t)/r$ per descrivere la rotazione della ruota.

$$T = \frac{1}{2} \cdot M_1 \cdot \dot{x}_1^2 + \frac{1}{2} \cdot M_2 \cdot \dot{x}_2^2 + \frac{1}{2} \cdot J \cdot \dot{\varphi}^2$$

e

$$U = M_1 \cdot g \cdot x_1 + M_2 \cdot g \cdot x_2$$

e infine

$$D = \frac{1}{2} \cdot f_v \cdot \dot{\varphi}^2$$

```
x: x(t);
x1: x;
x2: (1-x);
phi: x/r;

dx: diff(x, t);
ddx: diff(x, t, 2);

dx1: diff(x1, t);
dx2: diff(x2, t);
dphi: diff(phi, t);

T: 1/2*M1*dx1^2+1/2*M2*dx2^2+1/2*J*dphi^2;
U: M1*g*x1+M2*g*x2;
D: 1/2*f_v*dphi^2;
L: T-U;

eqn: diff(diff(L, dx), t)-diff(L, x)+diff(D, dx)=0;

sol: linsolve(eqn, ddx);
```

Il risultato ottenuto vale:

$$\frac{d^2}{dt^2} x(t) = \frac{g r^2 M_2}{r^2 M_2 + r^2 M_1 + J} - \frac{g r^2 M_1}{r^2 M_2 + r^2 M_1 + J} - \frac{f_v \left(\frac{d}{dt} x(t) \right)}{r^2 M_2 + r^2 M_1 + J}$$

La figura]2 mostra lo schema scicoslab per la simulazione.

Il plot della simulazione è mostrato nella figura 3.

4.3 Sistema meccanico smorzato

Determiniamo il modello matematico del sistema rappresentato nella figura 4.

Le equazioni dell'energia di questo sistema sono

$$U = \frac{1}{2} \cdot k \cdot x^2$$

$$T = \frac{1}{2} \cdot M \cdot \dot{x}^2$$

e

$$D = \frac{1}{2} \cdot f_v \cdot \dot{x}^2$$

Il seguente script di Maxima permette di ottenere il risultato desiderato:

```
x: x(t);
dx: diff(x, t);
ddx: diff(x, t, 2);

T: 1/2*M*dx^2;
```




Figura 2: Schema di simulazione della macchina di Atwood




Figura 3: Plot della simulazione della macchina di atwood




Figura 4: Rappresentazione di una massa oscillante con molla e ammortizzatore

```

U:1/2*k*x^2;
D:1/2*f_v*dx^2;
L:T-U;

eqn: diff(diff(L,dx),t)-diff(L,x) + diff(D,dx) = F(t);

sol: linsolve(eqn, ddx);

sol: expand(sol[1]);

```

che risulta essere

$$\frac{d^2}{dt^2} x(t) = -\frac{f_v \left(\frac{d}{dt} x(t) \right)}{M} + \frac{F(t)}{M} - \frac{k x(t)}{M}$$

La figura 5 mostra lo schema scicoslab di simulazione. Il risultato è mostrato nella figura 6.




Figura 5: Schema scicoslab del sistema meccanico




Figura 6: Plot della simulazione modelica (sopra) e equazione differenziale (sotto)

4.4 Ammortizzatore meccanico- Quarter Car

L'ammortizzatore meccanico da modellare è rappresentato nella figura 7.




Figura 7: Schema meccanico degli ammortizzatori di un veicolo (“quarter car”)

Le grandezze generalizzate del modello sono $z(t)$ e $z_t(t)$.

$$T = \frac{1}{2} \cdot M \cdot \dot{z}^2 + \frac{1}{2} \cdot m \cdot \dot{z}_t^2$$

$$U = M \cdot g \cdot z + m \cdot g \cdot z_t + \frac{1}{2} \cdot k \cdot (z - z_t)^2 + \frac{1}{2} \cdot k_t \cdot (z_t - z_r)^2$$

$$D = \frac{1}{2} \cdot d \cdot (\dot{z} - \dot{z}_t)^2 + \frac{1}{2} \cdot d_t \cdot (\dot{z}_t - \dot{z}_r)^2$$

```

z: z(t);
zt: zt(t);

dz: diff(z, t);
dzt: diff(zt, t);
ddz: diff(z, t, 2);
ddzt: diff(zt, t, 2);

T: 1/2*M*dz^2+1/2*m*dzt^2;
U: M*g*z+m*g*zt+1/2*k*(z-zt)^2+1/2*kt*(zt-zr)^2;
D: 1/2*d*(dz-dzt)^2 + 1/2*dt*(dzt-dzr)^2;
L: T-U;

eqn1: diff(diff(L, dz), t)-diff(L, z) + diff(D, dz) = 0;
eqn2: diff(diff(L, dzt), t)-diff(L, zt) + diff(D, dzt) = 0;

sol: linsolve([eqn1, eqn2], [ddz, ddzt]);
s1: rhs(sol[1]);
s2: rhs(sol[2]);
s1: ddz=expand(s1);

```

```
s2 : ddtz=expand(s2);
```

$$\frac{d^2}{dt^2} z(t) = \frac{d \cdot \dot{z}_t(t)}{M} - \frac{d \cdot \dot{z}(t)}{M} + \frac{k z_t(t)}{M} - \frac{k z(t)}{M} - g$$

$$\frac{d^2}{dt^2} z_t(t) = -\frac{d_t \cdot \dot{z}_t(t)}{m} - \frac{d \cdot \dot{z}_t(t)}{m} + \frac{d_t \cdot \dot{z}_r(t)}{m} + \frac{d \cdot \dot{z}(t)}{m} - \frac{k z_t(t)}{m} - \frac{k z(t)}{m} + \frac{k z_r(t)}{m} + \frac{k z(t)}{m} - g$$

4.5 Sistema Dischi-Molla rotanti

Determiniamo le equazioni differenziali del sistema della figura 8




Figura 8: Sistema masse e molle

Le grandezze in gioco sono Kt_1 : costante di momento del motore 1, Kt_2 costante di momento del motore 2, D_1 , D_2 e d sono le costanti di attrito del motore 1, motore2 e molla, mentre k è la costante della molla tra i due dischi. Inoltre J_1 e J_2 sono le inerzie di rotazione del sistema motore1+carico e motore2+carico.

$$T : \frac{1}{2} \cdot J_1 \cdot \dot{\varphi}_1^2 + \frac{1}{2} \cdot J_2 \cdot \dot{\varphi}_2^2$$

$$U : \frac{1}{2} \cdot k \cdot (\varphi_1 + \varphi_2)^2$$

$$D : \frac{1}{2} \cdot D_1 \cdot \dot{\varphi}_1^2 + \frac{1}{2} \cdot D_2 \cdot \dot{\varphi}_2^2 + \frac{1}{2} \cdot d \cdot (\dot{\varphi}_1 + \dot{\varphi}_2)^2$$

Il seguente script di Maxima determina le equazioni differenziali del sistema.

```
dphi1: diff(phi1, t);
dphi2: diff(phi2, t);
ddphi1: diff(phi1, t, 2);
ddphi2: diff(phi2, t, 2);

T:1/2*J1*dphi1^2 +1/2*J2*dphi2^2;
U:1/2*k*(phi1+phi2)^2;
D:1/2*D1*dphi1^2+1/2*D2*dphi2^2+1/2*d*(dphi1+dphi2)^2;

L:T-U;

eqn1: diff(diff(L, dphi1), t)-diff(L, phi1) + diff(D, dphi1) = Kt1*I1;
```

```

eqn2: diff(diff(L, dphi2), t) - diff(L, phi2) + diff(D, dphi2) = Kt2*I2;

sol: linsolve([eqn1, eqn2], [ddphi1, ddphi2]);

s1: expand(sol[1]);
s2: expand(sol[2]);

```

$$\frac{d^2}{dt^2} \varphi_1(t) = \frac{Kt_1 I_1}{J_1} - \frac{d}{dt} \varphi_1(t) \frac{D_1}{J_1} - \frac{d}{dt} \left(\frac{d}{dt} \varphi_2(t) \right) \frac{d}{J_1} - \frac{d}{dt} \left(\frac{d}{dt} \varphi_1(t) \right) \frac{k}{J_1} - \frac{k \varphi_2(t)}{J_1} - \frac{k \varphi_1(t)}{J_1}$$

$$\frac{d^2}{dt^2} \varphi_2(t) = \frac{Kt_2 I_2}{J_2} - \frac{d}{dt} \varphi_2(t) \frac{D_2}{J_2} - \frac{d}{dt} \left(\frac{d}{dt} \varphi_2(t) \right) \frac{d}{J_2} - \frac{d}{dt} \left(\frac{d}{dt} \varphi_1(t) \right) \frac{k}{J_2} - \frac{k \varphi_2(t)}{J_2} - \frac{k \varphi_1(t)}{J_2}$$

La figura]9 mostra lo schema scicoslab per la simulazione.




Figura 9: Schema di simulazione del sistema “dischi e molla”

Il plot della simulazione è mostrato nella figura 10.




Figura 10: Plot della simulazione del sistema “dischi e molla”

4.6 Sistema RC

Determiniamo ora il modello dinamico del sistema rappresentato nella figura 11.




Figura 11: Circuito elettrico con resistenze e capacità

Utilizziamo come grandezze generalizzate la carica q_1 che scorre nella maglia di sinistra e la carica q_2 che scorre in quella di destra.

$$U : \frac{1}{2 \cdot C_1} \cdot (q_1 - q_2)^2 + \frac{1}{2 \cdot C_2} \cdot q_2^2$$

$$T : 0$$

$$D : \frac{1}{2} \cdot R_1 \cdot \dot{q}_1^2 + \frac{1}{2} \cdot R_2 \cdot \dot{q}_2^2$$

Il programma di Maxima che risolve questo sistema è il seguente

```

q1: q1(t);
q2: q2(t);
dq1: diff(q1, t);
dq2: diff(q2, t);
ddq1: diff(q1, t, 2);
ddq2: diff(q2, t, 2);

U: 1/(2*C1)*(q1-q2)^2+1/(2*C2)*q2^2;
T: 0;
D: 1/2*R1*dq1^2+1/2*R2*dq2^2;
L: T-U;

eqn1: diff(diff(L, dq1), t)-diff(L, q1)+diff(D, dq1)=U0;
eqn2: diff(diff(L, dq2), t)-diff(L, q2)+diff(D, dq2)=0;

eqn1: expand(eqn1);
eqn2: expand(eqn2);

sol: linsolve([eqn1, eqn2], [dq1, dq2]);
s1: rhs(sol[1]);
s2: rhs(sol[2]);

dq1:=expand(s1);
dq2:=expand(s2);

```

La soluzione è rappresentata dalle equazioni seguenti

$$\frac{d}{dt} q_1(t) = \frac{U_0}{R_1} + \frac{q_2(t)}{C_1 R_1} - \frac{q_1(t)}{C_1 R_1}$$

$$\frac{d}{dt} q_2(t) = -\frac{q_2(t)}{C_2 R_2} - \frac{q_2(t)}{C_1 R_2} + \frac{q_1(t)}{C_1 R_2}$$

La figura]12 mostra lo schema scicoslab per la simulazione.




Figura 12: Schema di simulazione del sistema RC

Il plot della simulazione è mostrato nella figura 13.




Figura 13: Plot della simulazione del sistema RC

4.7 Sistema RCL

La figura seguente rappresenta un circuito RCL.

Scegliamo come grandezze generalizzate la carica Q_1 che entra nella resistenza da 2Ω , Q_2 quella del condensatore e Q_3 quella dell'induttanza. La somma delle tre cariche scorre nella resistenza da 4Ω .

$$U : \frac{1}{2 \cdot C_1} \cdot Q_3^2$$

$$T : \frac{1}{2} \cdot L_1 \cdot \dot{Q}_2^2$$

$$D : \frac{1}{2} \cdot R_1 \cdot \dot{Q}^2 + \frac{1}{2} \cdot R_2 \cdot \dot{Q}_1^2$$

Lo script di Maxima che determina le equazioni differenziali del sistema è il seguente




Figura 14: Sistema RCL

```

Q1: Q1(t);
Q2: Q2(t);
Q3: Q3(t);
Q: Q1+Q2+Q3;

dQ1: diff(Q1, t);
dQ2: diff(Q2, t);
dQ3: diff(Q3, t);
dQ: diff(Q, t);

ddQ1: diff(Q1, t, 2);
ddQ2: diff(Q2, t, 2);
ddQ3: diff(Q3, t, 2);
ddQ: diff(Q, t, 2);

U: 1/(2*C1)*Q3^2;
T: 1/2*L1*dQ2^2;
D: 1/2*R1*dQ^2+1/2*R2*dQ1^2;

L: T-U;

eqn1: diff(diff(L, dQ1), t)-diff(L, Q1) + diff(D, dQ1) = U0;
eqn2: diff(diff(L, dQ2), t)-diff(L, Q2) + diff(D, dQ2) = U0;
eqn3: diff(diff(L, dQ3), t)-diff(L, Q3) + diff(D, dQ3) = U0;

sol: linsolve([eqn1, eqn2, eqn3], [dQ1, ddQ2, dQ3]);

s1: rhs(sol[1]);
s2: rhs(sol[2]);
s3: rhs(sol[3]);

s1: dQ1=expand(s1);
s2: ddQ2=expand(s2);
s3: dQ3=expand(s3);

```

La soluzione è rappresentata da un sistema di 3 equazioni differenziali, di cui una di secondo ordine.

$$\frac{d}{dt} Q_1(t) = \frac{Q_3(t)}{C_1 R_2}$$

$$\frac{d^2}{dt^2} Q_2(t) = \frac{Q_3(t)}{C_1 L_1}$$

$$\frac{d}{dt} Q_3(t) = \frac{U_0}{R_1} - \frac{Q_3(t)}{C_1 R_2} - \frac{Q_3(t)}{C_1 R_1} - \frac{d}{dt} Q_2(t)$$

Il sistema contiene però unicamente due elementi che possono memorizzare energia, per cui potrà essere descritto tramite un sistema di equazioni differenziali di 2. ordine.

Una volta messo il sistema nello spazio degli stati, applicando una procedura di semplificazione (ad esempio *minss* in Scicoslab, si ottiene un sistema di 2. ordine.

La figura]15 mostra lo schema scicoslab per la simulazione.




Figura 15: Schema di simulazione del sistema RCL

Il plot della simulazione è mostrato nella figura 16.




Figura 16: Plot della simulazione del sistema RCL

4.8 Motore elettrico DC con carico

La figura 17 mostra lo schema di un motore DC con una tensione in entrata e_a . Le grandezze associate a questo motore che interessano la modellazione sono:

L'inerzia di rotazione $J = J_{motore} + J_{carico}$

Le costanti del motore K_t (costante di coppia) e K_b (costante di back emf), la resistenza di armatura R_a e l'induttanza di armatura L_a .

$$T : 1/2 \cdot L \cdot \dot{q}^2 + 1/2 \cdot J \cdot \dot{\varphi}^2$$




Figura 17: DC motor

$$U : 0$$

$$D : 1/2 \cdot R \cdot \dot{q}^2 + 1/2 \cdot f_v \cdot \dot{\varphi}^2$$

Il seguente script di Maxima calcola le equazioni differenziali associate al motore utilizzando come grandezze generalizzate la carica Q nel circuito elettrico e l'angolo φ nella parte meccanica.

```

Q:Q(t);
phi:phi(t);

dQ:diff(Q,t);
dphi:diff(phi,t);
ddQ:diff(Q,t,2);
ddphi:diff(phi,t,2);

T:1/2*L*dQ^2+1/2*J*dphi^2;
U:0;
D:1/2*R*dQ^2+1/2*f_v*dphi^2;
L:T-U;

eqn1:diff(diff(L,dQ),t)-diff(L,Q)+diff(D,dQ)=U0-Kb*dphi;
eqn2:diff(diff(L,dphi),t)-diff(L,phi)+diff(D,dphi)=Kt*dQ;

sol:linsolve([eqn1,eqn2],[ddQ,ddphi]);
s1:rhs(sol[1]);
s2:rhs(sol[2]);
s1:expand(s1);
s2:expand(s2);

```

La soluzione porta al seguente sistema di equazioni differenziali

$$\frac{d^2}{dt^2} Q(t) = \frac{U_0}{L} - \frac{\frac{d}{dt} Q(t) R}{L} - \frac{Kb \left(\frac{d}{dt} \varphi(t) \right)}{L}$$

$$\frac{d^2}{dt^2} \varphi(t) = \frac{Kt \left(\frac{d}{dt} Q(t) \right)}{J} - \frac{f_v \left(\frac{d}{dt} \varphi(t) \right)}{J}$$

Tenendo conto che la carica Q entra in gioco solo come prima e seconda derivata possiamo sostituire $dQ/dt = I$ e otteniamo il seguente sistema

$$\frac{d}{dt} I(t) = \frac{U_0}{L} - \frac{I(t) R}{L} - \frac{Kb}{L} \left(\frac{d}{dt} \varphi(t) \right)$$

$$\frac{d^2}{dt^2} \varphi(t) = \frac{Kt I(t)}{J} - \frac{fv}{J} \left(\frac{d}{dt} \varphi(t) \right)$$

La figura]18 mostra lo schema scicoslab per la simulazione.




Figura 18: Schema di simulazione del motore DC

Il plot della simulazione è mostrato nella figura 19.




Figura 19: Plot della simulazione del motore DC

4.9 Sospensione magnetica

$$T : 1/2 \cdot L \cdot \dot{Q}^2 + 1/2 \cdot L \cdot \dot{z}^2$$

$$U : m \cdot g \cdot z$$




Figura 20: Sospensione magnetica

$$D : 1/2 \cdot R \cdot \dot{Q}^2$$

Le grandezze generalizzate utilizzate sono le cariche nella parte elettrica ($Q(t)$) e la posizione $z(t)$ della boccia.

```

Q:Q(t);
z:z(t);

dQ:diff(Q,t);
dz:diff(z,t);
ddQ:diff(Q,t,2);
ddz:diff(z,t,2);

T:1/2*L*dQ^2+1/2*L*dz^2;
U:m*g*z;
D:1/2*R*dQ^2;
L:T-U;

eqn1:diff(diff(L,dQ),t)-diff(L,Q)+diff(D,dQ)=U0;
eqn2:diff(diff(L,dz),t)-diff(L,z)=K*dQ^2/z^2;

sol:linsolve([eqn1,eqn2],[ddQ,ddz]);
s1:expand(sol[1]);
s2:expand(sol[2]);

```

Le equazioni differenziali risultanti sono:

$$\frac{d^2}{dt^2} Q(t) = \frac{U_0}{L} - \frac{\frac{d}{dt} Q(t) R}{L}$$

$$\frac{d^2}{dt^2} z(t) = \frac{\left(\frac{d}{dt} Q(t)\right)^2 K}{z^2(t) L} - \frac{g m}{L}$$

A questo punto si può sostituire la carica $Q(t)$ con la corrente $i(t)$ e il sistema diventa di 3. ordine.

4.10 Pendolo inverso

Modelliamo il sistema carrello+pendolo inverso della figura 21. Scegliamo come variabili generalizzate le grandezze $x(t)$ e $\varphi(t)$.




Figura 21: Pendolo inverso

$$T : \frac{1}{2} \cdot M \cdot \dot{x}_1^2 + \frac{1}{2} \cdot m \cdot (\dot{x}_2^2 + \dot{x}_3^2) + \frac{1}{2} \cdot \text{Theta} \cdot \dot{\varphi}^2$$

$$U : m \cdot g \cdot r \cdot \cos(\varphi)$$

$$D : \frac{1}{2} \cdot d \cdot \dot{x}_1^2$$

Lo script di Maxima che calcola il sistema è il seguente:

```
x:x(t);
phi:phi(t);

dx:diff(x,t);
dphi:diff(phi,t);
ddx:diff(x,t,2);
ddphi:diff(phi,t,2);

x1:x;
x2:x-r*sin(phi);
x3:r*cos(phi);
dx1:diff(x1,t);
dx2:diff(x2,t);
dx3:diff(x3,t);

T:1/2*M*dx1^2+1/2*m*(dx2^2+dx3^2)+1/2*Theta*dphi^2;
U:m*g*r*cos(phi);
D:1/2*d*dx^2;
L:T-U;

eqn1:diff(diff(L,dphi),t)-diff(L,phi)+diff(D,dphi)=0;
eqn2:diff(diff(L,dx),t)-diff(L,x)+diff(D,dx)=F;

eqn1:expand(eqn1);
eqn2:expand(eqn2);
eqn1:trigreduce(eqn1);
eqn2:trigreduce(eqn2);
eqn1:trigsimp(eqn1);
eqn2:trigsimp(eqn2);
```

```
sol : linsolve ([ eqn1 , eqn2 ] , [ ddx , ddphi ] );
s1 : rhs ( sol [ 1 ] );
s2 : rhs ( sol [ 2 ] );
```

Il risultato ottenuto è il seguente

$$\frac{d^2}{dt^2} x(t) = -\frac{\Theta (-F + d\dot{x} + m r \sin \varphi(t) \dot{\varphi}^2) + m r^2 (d\dot{x} - F) + \sin \varphi(t) (m^2 r^3 \dot{\varphi}^2 - g m^2 r^2 \cos \varphi(t))}{r^2 (m M + m^2) + \Theta (M + m) - m^2 r^2 \cos^2 \varphi(t)}$$

$$\frac{d^2}{dt^2} \varphi(t) = -\frac{\sin \varphi(t) (r (-g m M - g m^2) + m^2 r^2 \cos \varphi(t) \dot{\varphi}^2) + m r \cos \varphi(t) (d\dot{x} - F)}{r^2 (m M + m^2) + \Theta (M + m) - m^2 r^2 \cos^2 \varphi(t)}$$

4.11 Pendolo di Furuta

Modelliamo il pendolo di Furuta della figura 22.




Figura 22: Pendolo di Furuta

I parametri del pendolo da considerare sono la lunghezza del braccio sul motore L_0 , con la sua inerzia J_0 e il suo attrito R_0 , la distanza tra base e centro di massa del braccio mobile L_1 con la sua inerzia J_1 e il suo attrito R_1 .

Le variabili generalizzate sono i due angoli φ_0 (rotazione del braccio del motore) e φ_1 (rotazione del pendolo).

Le grandezze da calcolare per la lagrangiana sono le seguenti:

$$T = \frac{1}{2} \cdot (J_0 \cdot \dot{\varphi}_0^2 + J_1 \cdot \dot{\varphi}_1^2 + m \cdot (\dot{x}_1^2 + \dot{y}_1^2 + \dot{z}_1^2))$$

con \dot{x}_1 , \dot{y}_1 e \dot{z}_1 sono le velocità del centro di massa (CM) del pendolo.

$$U = m \cdot z_1 \cdot g$$

$$D = \frac{1}{2} \cdot (R_0 \cdot \dot{\varphi}_0^2 + R_1 \cdot \dot{\varphi}_1^2)$$

Questo è lo script di Maxima che permette di ottenere le equazioni differenziali finali.

```

phi0: phi0(t);
phi1: phi1(t);
I: I(t);

dphi0: diff(phi0, t);
dphi1: diff(phi1, t);
ddphi0: diff(phi0, t, 2);
ddphi1: diff(phi1, t, 2);

x1: L0*sin(phi0)+L1*sin(phi1)*cos(phi0);
y1: L0*cos(phi0)-L1*sin(phi1)*sin(phi0);
z1: L1*cos(phi1);

dx1: diff(x1, t);
dy1: diff(y1, t);
dz1: diff(z1, t);

T: 1/2*(J0*dphi0^2+J1*dphi1^2+m*(dx1^2+dy1^2+dz1^2));
U: m*z1*g;
D: 1/2*(R0*dphi0^2+R1*dphi1^2);

L: T-U;

eqn1: diff(diff(L, dphi0), t)-diff(L, phi0)+diff(D, dphi0)=Kt*I;
eqn2: diff(diff(L, dphi1), t)-diff(L, phi1)+diff(D, dphi1)=0;

sol: linsolve([eqn1, eqn2], [ddphi0, ddphi1]);

s1: trigsimp(sol[1]);
s2: trigsimp(sol[2]);

```

Dopo la linearizzazione si ottengono le matrici seguenti

$$A = \begin{pmatrix} 0 & \frac{1}{m J_0 L_1^2 + m J_1 L_0^2 + J_0 J_1} (-m L_1^2 - J_1) R_0 & 0 & 0 \\ 0 & 0 & -\frac{g m^2 L_0 L_1^2}{m J_0 L_1^2 + m J_1 L_0^2 + J_0 J_1} & \frac{m L_0 L_1 R_1}{m J_0 L_1^2 + m J_1 L_0^2 + J_0 J_1} \\ 0 & 0 & 0 & \frac{1}{m J_0 L_1^2 + m J_1 L_0^2 + J_0 J_1} (m L_0^2 + J_0) R_1 \\ 0 & \frac{m L_0 L_1 R_0}{m J_0 L_1^2 + m J_1 L_0^2 + J_0 J_1} & \frac{(g m^2 L_0^2 + g m J_0) L_1}{m J_0 L_1^2 + m J_1 L_0^2 + J_0 J_1} & -\frac{(m L_0^2 + J_0) R_1}{m J_0 L_1^2 + m J_1 L_0^2 + J_0 J_1} \end{pmatrix}$$

$$B = \begin{pmatrix} 0 \\ \frac{K t (m L_1^2 + J_1)}{m J_0 L_1^2 + m J_1 L_0^2 + J_0 J_1} \\ 0 \\ -\frac{K t m L_0 L_1}{m J_0 L_1^2 + m J_1 L_0^2 + J_0 J_1} \end{pmatrix}$$

$$C = \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \end{pmatrix}$$

$$D = \begin{pmatrix} 0 \\ 0 \end{pmatrix}$$